

1 Introduction

Il existe deux méthodes **SPICE** pour modéliser un transformateur :

- par couplage d'inductances ;
- par un circuit équivalent.

La première méthode nécessite de connaître les valeurs des inductances du primaire et du secondaire. Cependant, ces valeurs ne sont pas toujours les grandeurs d'intérêt (par exemple, dans le cas du *flyback* ou du transformateur idéal).

La seconde méthode permet de réaliser un transformateur *idéal*. Les écarts à l'idéalité peuvent alors être modélisés par des éléments *circuit* supplémentaires (notamment l'inductance magnétisante, grandeur d'intérêt du *flyback*).

2 Modèle équivalent

Soit le transformateur *idéal* de rapport $1 : n$ de la figure 1.




FIGURE 1 – Un transformateur idéal

Les équations gouvernant ce quadripôle sont :

$$\begin{cases} V_2 = n V_1 \\ I_1 = n I_2 . \end{cases}$$

Il est alors possible de réaliser un circuit, illustré à la figure 2, satisfaisant ces équations.




FIGURE 2 – Modèle équivalent d'un transformateur idéal

Ce circuit est réalisable dans **SPICE**, et implémente le même comportement qu'un transformateur idéal.

3 Sources de courant et de tension

Le modèle de la figure 2, bien que correct conceptuellement, présente un grand défaut : il n'est pas possible de mettre les bornes 1 et 2 en circuit ouvert, ni de mettre les bornes 3 et 4 en court-circuit. En effet, une source idéale de courant ne peut être en circuit ouvert. De même, une source idéale de tension ne peut être en court-circuit.

Pour palier à ce problème, il est nécessaire de placer une résistance de grande valeur en *parallèle* avec la source de courant, ainsi qu'une résistance de faible valeur en *série* avec la source de tension. La figure 3 illustre ce raisonnement ¹.




FIGURE 3 – Solution au problème des sources idéales

4 Implémentation SPICE

Nous disposons maintenant du modèle équivalent du transformateur. Il ne nous reste plus qu'à l'implémenter dans SPICE. Pour ce faire, nous avons besoin de sources commandées.

4.1 Sources de tension commandées en tension

Les sources de tension commandées en tension respectent la syntaxe suivante.

Source de tension commandée en tension

E<Nom> N+ N- NC+ NC- Valeur

Où :

- Nom est le nom de la source ;
- N+ est le nœud de sortie positif ;
- N- est le nœud de sortie négatif ;
- NC+ est le nœud de contrôle positif ;
- NC- est le nœud de contrôle négatif ;
- Valeur est le coefficient liant la tension de sortie à la tension de contrôle :

$$V_{N+} - V_{N-} = \text{Valeur} \times (V_{NC+} - V_{NC-}).$$

1. Les valeurs de 1 GΩ et de 1 nΩ ont été choisies arbitrairement. Il serait parfaitement légitime de prendre une autre paire de valeurs.

4.2 Sources de courant commandées en courant

Les sources de courant commandées en courant respectent la syntaxe suivante.

Source de courant commandée en courant

F<Nom> N+ N- VSens Valeur

Où :

- Nom est le nom de la source ;
- N+ est le nœud de sortie positif (le courant *sort* par cette borne) ;
- N- est le nœud de sortie négatif (le courant *rentre* par cette borne) ;
- VSens est le *nom* d'une source de *tension* dans laquelle le courant de contrôle est mesuré ;
- Valeur est le coefficient liant le courant de sortie au courant de contrôle :

$$I_{N+ \rightarrow N-} = \text{Valeur} \times I_{\text{VSens}}.$$

Notons que la source de tension VSens, utilisée par la source de courant, peut ne *pas imposer de différence de potentielle*. Ainsi, la source VSens peut être vue comme un simple *ampèremètre*.

4.3 Modèle SPICE

Nous disposons à présent de tous les éléments nécessaire pour réaliser un transformateur idéal de rapport 1 : n. Le schéma du code ci-dessous est repris à la figure 4.

Modèle SPICE d'un transformateur idéal de rapport 1:N

```
.SUBCKT XFRM 1 2 3 4 PARAMS: N = 1
EOut 5 4 1 2 {N}
FIn 2 1 VAm {N}
VAm 6 3 0
RIn 1 2 1G
ROut 5 6 1n
.ENDS XFRM
```




FIGURE 4 – Schéma du modèle SPICE